

# NAG Library Routine Document

## F07FDF (DPOTRF)

**Note:** before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

### 1 Purpose

F07FDF (DPOTRF) computes the Cholesky factorization of a real symmetric positive definite matrix.

### 2 Specification

```
SUBROUTINE F07FDF (UPLO, N, A, LDA, INFO)
  INTEGER          N, LDA, INFO
  REAL (KIND=nag_wp) A(LDA,*)
  CHARACTER(1)    UPLO
```

The routine may be called by its LAPACK name *dpotrf*.

### 3 Description

F07FDF (DPOTRF) forms the Cholesky factorization of a real symmetric positive definite matrix  $A$  either as  $A = U^T U$  if  $UPLO = 'U'$  or  $A = LL^T$  if  $UPLO = 'L'$ , where  $U$  is an upper triangular matrix and  $L$  is lower triangular.

### 4 References

Demmel J W (1989) On floating-point errors in Cholesky *LAPACK Working Note No. 14* University of Tennessee, Knoxville <http://www.netlib.org/lapack/lawnspdf/lawn14.pdf>

Golub G H and Van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

### 5 Parameters

1: UPLO – CHARACTER(1) *Input*

*On entry:* specifies whether the upper or lower triangular part of  $A$  is stored and how  $A$  is to be factorized.

UPLO = 'U'

The upper triangular part of  $A$  is stored and  $A$  is factorized as  $U^T U$ , where  $U$  is upper triangular.

UPLO = 'L'

The lower triangular part of  $A$  is stored and  $A$  is factorized as  $LL^T$ , where  $L$  is lower triangular.

*Constraint:* UPLO = 'U' or 'L'.

2: N – INTEGER *Input*

*On entry:*  $n$ , the order of the matrix  $A$ .

*Constraint:*  $N \geq 0$ .

3: A(LDA,\*) – REAL (KIND=nag\_wp) array Input/Output

**Note:** the second dimension of the array A must be at least  $\max(1, N)$ .

*On entry:* the  $n$  by  $n$  symmetric positive definite matrix  $A$ .

If UPLO = 'U', the upper triangular part of  $A$  must be stored and the elements of the array below the diagonal are not referenced.

If UPLO = 'L', the lower triangular part of  $A$  must be stored and the elements of the array above the diagonal are not referenced.

*On exit:* the upper or lower triangle of  $A$  is overwritten by the Cholesky factor  $U$  or  $L$  as specified by UPLO.

4: LDA – INTEGER Input

*On entry:* the first dimension of the array A as declared in the (sub)program from which F07FDF (DPOTRF) is called.

*Constraint:*  $LDA \geq \max(1, N)$ .

5: INFO – INTEGER Output

*On exit:* INFO = 0 unless the routine detects an error (see Section 6).

## 6 Error Indicators and Warnings

INFO < 0

If INFO =  $-i$ , argument  $i$  had an illegal value. An explanatory message is output, and execution of the program is terminated.

INFO > 0

The leading minor of order  $\langle value \rangle$  is not positive definite and the factorization could not be completed. Hence  $A$  itself is not positive definite. This may indicate an error in forming the matrix  $A$ . To factorize a symmetric matrix which is not positive definite, call F07MDF (DSYTRF) instead.

## 7 Accuracy

If UPLO = 'U', the computed factor  $U$  is the exact factor of a perturbed matrix  $A + E$ , where

$$|E| \leq c(n)\epsilon|U^T||U|,$$

$c(n)$  is a modest linear function of  $n$ , and  $\epsilon$  is the *machine precision*. If UPLO = 'L', a similar statement holds for the computed factor  $L$ . It follows that  $|e_{ij}| \leq c(n)\epsilon\sqrt{a_{ii}a_{jj}}$ .

## 8 Parallelism and Performance

F07FDF (DPOTRF) is threaded by NAG for parallel execution in multithreaded implementations of the NAG Library.

F07FDF (DPOTRF) makes calls to BLAS and/or LAPACK routines, which may be threaded within the vendor library used by this implementation. Consult the documentation for the vendor library for further information.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the OpenMP environment used within this routine. Please also consult the Users' Note for your implementation for any additional implementation-specific information.

## 9 Further Comments

The total number of floating-point operations is approximately  $\frac{1}{3}n^3$ .

A call to F07FDF (DPOTRF) may be followed by calls to the routines:

F07FEF (DPOTRS) to solve  $AX = B$ ;

F07FGF (DPOCON) to estimate the condition number of  $A$ ;

F07FJF (DPOTRI) to compute the inverse of  $A$ .

The complex analogue of this routine is F07FRF (ZPOTRF).

## 10 Example

This example computes the Cholesky factorization of the matrix  $A$ , where

$$A = \begin{pmatrix} 4.16 & -3.12 & 0.56 & -0.10 \\ -3.12 & 5.03 & -0.83 & 1.18 \\ 0.56 & -0.83 & 0.76 & 0.34 \\ -0.10 & 1.18 & 0.34 & 1.18 \end{pmatrix}.$$

### 10.1 Program Text

```

Program f07fdfe

!      F07FDF Example Program Text

!      Mark 25 Release. NAG Copyright 2014.

!      .. Use Statements ..
      Use nag_library, Only: dpotrf, nag_wp, x04caf
!      .. Implicit None Statement ..
      Implicit None
!      .. Parameters ..
      Integer, Parameter          :: nin = 5, nout = 6
!      .. Local Scalars ..
      Integer                     :: i, ifail, info, lda, n
      Character (1)               :: uplo
!      .. Local Arrays ..
      Real (Kind=nag_wp), Allocatable :: a(:, :)
!      .. Executable Statements ..
      Write (nout,*) 'F07FDF Example Program Results'
!      Skip heading in data file
      Read (nin,*)
      Read (nin,*) n
      lda = n
      Allocate (a(lda,n))

!      Read A from data file

      Read (nin,*) uplo
      If (uplo=='U') Then
         Read (nin,*)(a(i,i:n),i=1,n)
      Else If (uplo=='L') Then
         Read (nin,*)(a(i,1:i),i=1,n)
      End If

!      Factorize A
!      The NAG name equivalent of dpotrf is f07fdf
      Call dpotrf(uplo,n,a,lda,info)

      Write (nout,*)
      Flush (nout)
      If (info==0) Then

!      Print factor

```

```

!      ifail: behaviour on error exit
!      =0 for hard exit, =1 for quiet-soft, =-1 for noisy-soft
      ifail = 0
      Call x04caf(uplo,'Nonunit',n,n,a,lda,'Factor',ifail)

      Else
        Write (nout,*) 'A is not positive definite'
      End If

      End Program f07fdfe

```

## 10.2 Program Data

```

F07FDF Example Program Data
  4                               :Value of N
  'L'                             :Value of UPLO
  4.16
 -3.12   5.03
  0.56  -0.83   0.76
 -0.10   1.18   0.34   1.18   :End of matrix A

```

## 10.3 Program Results

F07FDF Example Program Results

Factor	1	2	3	4
1	2.0396			
2	-1.5297	1.6401		
3	0.2746	-0.2500	0.7887	
4	-0.0490	0.6737	0.6617	0.5347

---